

Appendix C

Mplus Input

Mplus Syntax

```

1  MODEL:      M on Mpre Ypre;
2              Y on Ypre M Mpre;
3              Mpre WITH Ypre;
4              [Mpre ];
5              [Ypre ];
6              [M ];
7              [Y ];
8
9  MODEL 0: !Regressions in group 1
10          M on Mpre (sMZG1)
11              Ypre (sMWG1);
12          Y on Ypre (sYWG1)
13              M (sYMG1)
14              Mpre (sYZG1);
15
16          !Covariance between Mpre and Ypre in group 1
17          Mpre WITH Ypre ;
18
19          !Conditional means of Mpre and Ypre in group 1
20          [Mpre ] (mZG1);
21          [Ypre ] (mWG1);
22
23          !Intercepts
24          [M ] (iMG1);
25          [Y ] (iYG1);
26
27  MODEL 1: !Regressions in Group 2
28          M on Mpre (sMZG2)
29              Ypre (sMWG2);
30          Y on Ypre (sYWG2)
31              M (sYMG2)
32              Mpre (sYZG2);
33
34          !Covariance between Z and W in group 2
35          Mpre WITH Ypre ;
36
37          !Conditional means of Z and W in group 2
38          [Mpre ] (mZG2);
39          [Ypre ] (mWG2);
40
41          !Intercepts
42          [M ] (iMG2);
43          [Y ] (iYG2);
44
45  MODEL CONSTRAINT:
46
47  !!!!!!!!!!!!!!! New Parameters !!!!!!!!!!!!!!!

```

```

48
49 !Adjusted means
50 NEW(mYadjG1); !adjusted mean in control group
51 NEW(mYadjG2); !adjusted mean in treatment group
52
53 !Unconditional means of Mpre (Z), Ypre (W) and M (M)
54 NEW(mZ);
55 NEW(mW);
56 NEW(mM);
57
58 !Group specific means of M
59 NEW(mMG1);
60 NEW(mMG2);
61
62 !Intercepts of the regressions Y on Mpre and Ypre
63 NEW(iYrG1);
64 NEW(iYrG2);
65
66 !Slopes of the regressions Y on Mpre and Ypre
67 NEW(sYZrG1);
68 NEW(sYWrG1);
69 NEW(sYZrG2);
70 NEW(sYWrG2);
71
72 !Adjusted means for the group specific regression Y on Mpre and Ypre
73 !used for computing the total effect
74 NEW(mYadtG1);
75 NEW(mYadtG2);
76
77 !Effects
78 NEW(direct*10);    !direct effect of X on Y
79 NEW(total*20);    !total effect
80 NEW(indirect*10); !indirect effect
81
82 NEW(nde0 nde1, adet);
83 NEW(mMadjG1, mMadjG2); ! adjusted means of M
84
85 NEW(ga10, ga11, ga12, ga13);
86
87 !!!!!!!!!!!!!!!!!!!!! Calculations !!!!!!!!!!!!!!!!!!!!!
88
89 !Unconditional means of Mpre, Ypre and M
90 mZ = (1006*mZG1 + 994*mZG2)/2000;
91 mW = (1006*mWG1 + 994*mWG2)/2000;
92 mM = (1006*mMG1 + 994*mMG2)/2000;
93
94 !Conditional means of M
95 mMG1 = iMG1 + sMZG1*mZG1 + sMWG1*mWG1;
96 mMG2 = iMG2 + sMZG2*mZG2 + sMWG2*mWG2;
97
98 !Adjusted means of M

```

```

99  mMadjG1 = iMG1 + sMZG1*mZ + sMWG1*mW;
100 mMadjG2 = iMG2 + sMZG2*mZ + sMWG2*mW;
101
102 !Adjusted means of the outcome variable Y in both groups
103 mYadjG1 = iYG1 + sYWG1*mW + sYZG1*mZ + sYMG1*mM;
104 mYadjG2 = iYG2 + sYWG2*mW + sYZG2*mZ + sYMG2*mM;
105
106 !Group specific regressions Y on Mpre and Ypre
107 iYrG1 = iYG1 + sYMG1*iMG1;
108 iYrG2 = iYG2 + sYMG2*iMG2;
109
110 sYZrG1 = sYZG1 + sYMG1*sMZG1;
111 sYZrG2 = sYZG2 + sYMG2*sMZG2;
112
113 sYWrG1 = sYWG1 + sYMG1*sMWG1;
114 sYWrG2 = sYWG2 + sYMG2*sMWG2;
115
116 !Adjusted means for the computation of the total effect
117 !based on group specific regressions Y on Mpre and Ypre
118 mYadtG1 = iYrG1 + sYWrG1*mW + sYZrG1*mZ ;
119 mYadtG2 = iYrG2 + sYWrG2*mW + sYZrG2*mZ ;
120
121 !Effects
122 direct = mYadjG2 - mYadjG1;
123 total = mYadtG2 - mYadtG1;
124 indirect = total - direct;
125

```

Detailed Explanation of the Mplus Syntax

In the MODEL section, we specify the four regression models shown in Equations 13-16 (Appendix A, main article). In lines 10-14 and 28-32 the group-specific regression coefficients of $E^{X=x}(M | M^{pre}, Y^{pre})$ and $E^{X=x}(Y | M^{pre}, Y^{pre}, M)$ are specified and named. The corresponding intercepts for these regressions appear in lines 24-25 and 42-43 and the conditional expectations $E^{X=x}(M^{pre})$ and $E^{X=x}(Y^{pre})$ are defined in lines 20-21 and 38-39. Names for model parameters and derived quantities are specified using parentheses in *Mplus*. A complete list of parameter names and the corresponding indicated quantities is provided in Table C1.

In the MODEL CONSTRAINT section of the *Mplus* input file, we first introduce necessary parameters for the computation of causal effects. Derived parameters appear as named constraints in the *Mplus* input file. In particular, the new parameters are the unconditional expectations $E(M^{pre})$, $E(Y^{pre})$ and $E(M)$, the conditional expectations $E(M|X=x)$ of both groups and the coefficients of the two conditional regressions $E^{X=x}(Y | M^{pre}, Y^{pre})$. We define new parameters in input lines 50-85 and compute them in lines 90-124. For example, calculation of unconditional expectations of M^{pre} , Y^{pre} and M as shown in Equation 18 is specified in lines 90-96.

In line 103 and 104 of the *Mplus* input file, we compute unconditional expectations of true-outcome variables w.r.t the direct effect. We compute the group-specific regression coefficients and the intercepts of $E^{X=x}(Y | M^{pre}, Y^{pre})$ in lines 107-114 and the expectations of true-outcome variables w.r.t the total effect in lines 118-119.

The average total, direct and indirect effects are also NEW PARAMETERS defined as functions of the parameters previously specified. The average direct effect is the difference $E(\tau_{1,t_M}) - E(\tau_{0,t_M})$ (see line 122). The average total effect is computed in line 123 as the difference between the two adjusted means with respect to total effects. The average indirect effect is the difference between the average total effect and the average direct effect $E(\delta_{10}) - E(\delta_{10,t_M})$ (see line 124).

Table C1

List of constraint names used in the Mplus input file.

Constraint name	Addressed Quantity	Constraint name	Addressed Quantity
sMZG1	β_{01}	Unconditional and conditional expectations	
sMWG1	β_{02}	mZ	$E(M^{pre})$
sYWG1	γ_{03}	mW	$E(Y^{pre})$
sYMG1	γ_{02}	mM	$E(M)$
sYZG1	γ_{01}	mMG1	$E(M \mid X=0)$
mZG1	$E(Z \mid X=0)$	mMG2	$E(M \mid X=1)$
mWG1	$E(W \mid X=0)$	Coefficients of $E^{X=0}(Y \mid M^{pre}, Y^{pre})$	
iMG1	β_{00}	iYrG1	α_{00}
iYG1	γ_{00}	sYZrG1	α_{01}
sMZG2	β_{11}	sYWrG1	α_{02}
sMWG2	β_{12}	Coefficients of $E^{X=1}(Y \mid M^{pre}, Y^{pre})$	
sYWG2	γ_{13}	iYrG2	α_{10}
sYMG2	γ_{12}	sYZrG2	α_{11}
sYZG2	γ_{11}	sYWrG2	α_{12}
mZG2	$E(M^{pre} \mid X=1)$	Adjusted means w.r.t. total effects	
mWG2	$E(Y^{pre} \mid X=1)$	mYadtG1	$E(\tau_0)$
iMG2	β_{10}	mYadtG2	$E(\tau_1)$
iYG2	γ_{10}	Effects	
Adjusted means w.r.t. direct effects		direct	$E(\delta_{10, t_M})$
mYadjG1	$E(\tau_0, t_M)$	total	$E(\delta_{10})$
mYadjG2	$E(\tau_1, t_M)$	indirect	$E(\delta_{10}) - E(\delta_{10, t_M})$

Note. See Equations 13-17 for details on regression coefficients α , β and γ .

Appendix D

Mplus Input for Data Generation

```

1  MONTECARLO:
2      NAMES ARE Ypre Mpre Y M;
3      NOBSERVATIONS = 1006 994;
4      NREPS = 1;
5      SEED = 53483;
6      NGROUPS=2;
7      SAVE = mediat91.dat;
8  ANALYSIS:
9      ESTIMATOR = ML;
10
11  MODEL MONTECARLO:
12
13  MODEL POPULATION:
14      M on Mpre*.80 Ypre*.00 ;
15      Y on Ypre*.90 M *.50 Mpre*.00 ;
16
17  !Control group
18  MODEL POPULATION-G1:
19      M on Mpre*.80 Ypre*.00 ;

```

```

20      Y on Ypre*.90 M *.50 Mpre*.00 ;
21      Mpre WITH Ypre*85;
22      Mpre*100;
23      Ypre*100;
24      M*25;
25      Y*16;
26      [Mpre*100];
27      [Ypre*100];
28
29  !Treatment group
30  MODEL POPULATION-G2:
31      M on Mpre*.80 Ypre*.00 ;
32      Y on Ypre*.90 M *.50 Mpre*.00 ;
33      Mpre WITH Ypre*85;
34      Mpre*100;
35      Ypre*100;
36      M*25;
37      Y*16;
38      [Mpre*100];
39      [Ypre*100];
40      [M*20];
41      [Y*10];

```